



# ***Proc R di WPS***

## ***guida dell'utente e riferimento***

***Programmazione nei linguaggi di SAS e R***



Versione: 4.0.3

Copyright © 2002-2018 World Programming Limited

[www.worldprogramming.com](http://www.worldprogramming.com)

# Indice

<b>Introduzione.....</b>	<b>3</b>
<b>Avvio rapido.....</b>	<b>5</b>
<b>Installazione e configurazione.....</b>	<b>9</b>
Impostazione della variabile di ambiente <i>R_HOME</i> .....	9
Impostazione di <i>R_HOME</i> tramite l'istruzione <code>OPTIONS</code> (tutti i sistemi operativi).....	10
Impostazione di <i>R_HOME</i> tramite una variabile di ambiente di avvio del server WPS (tutti i sistemi operativi).....	11
Impostazione di <i>R_HOME</i> in uno script del profilo shell UNIX.....	12
Note specifiche della piattaforma.....	12
Windows.....	12
Piattaforme UNIX o Linux.....	13
macOS.....	14
<b>Funzionamento di base.....</b>	<b>15</b>
Test dell'installazione e configurazione di WPS e R.....	15
Trasferimento dei dati tra WPS ed R.....	16
Uso dell'istruzione <code>EXPORT</code> .....	16
Uso dell'istruzione <code>IMPORT</code> .....	17
Uso della grafica di R.....	17
Uso di ulteriori pacchetti di R.....	19
<b>Riferimento.....</b>	<b>21</b>
Come leggere gli diagrammi sintattici.....	21
Procedura R.....	23
PROC R.....	23
ASSIGN.....	24
EXECUTE.....	26
EXPORT.....	26
IMPORT.....	28
LOAD.....	31
SAVE.....	32
SUBMIT.....	33
<b>Ulteriore lettura.....</b>	<b>36</b>
<b>Avvisi legali.....</b>	<b>37</b>

# Introduzione

La procedura R consente ai programmi scritti in linguaggio di SAS di includere il codice scritto in linguaggio R.

Questo documento fornisce le informazioni necessarie per configurare e utilizzare la procedura R (PROC R) dall'interno di WPS.

## Motivazione


Il linguaggio R è diffuso in segmenti significativi della comunità di analisi dei dati per i seguenti motivi:

- Ha stabilito una forte posizione all'interno dell'ecosistema open source.
- Per il linguaggio, è disponibile una grande quantità di funzionalità statistiche.
- Può offrire alcune funzionalità statistiche non disponibili in WPS.
- Vi sono già molti programmatori R che desiderano utilizzare queste competenze.

L'unione del linguaggio R e SAS crea una soluzione in cui la totalità è maggiore della somma dei singoli componenti. Tale interoperabilità consente la scrittura della mole di elaborazione e analisi dati nel robusto linguaggio di SAS dalle elevate prestazioni, utilizzando al contempo le nuove e specialistiche funzionalità statistiche, presenti nel mondo di R.

## Il quadro generale

Si consiglia di codificare i programmi principalmente in linguaggio di SAS, ricorrendo a R laddove sono necessarie statistiche specialistiche:



È semplice trasferire i dati rapidamente ed efficacemente tra l'ambiente WPS e R utilizzando la sintassi di `PROC R`. Una volta trasferiti i dati, la normale sintassi di R viene utilizzata per scrivere un programma R esattamente nello stesso modo in cui si scrive un programma R in un ambiente non WPS.

Una volta completato un programma R, è possibile trasferire facilmente i dati all'ambiente WPS per continuare con l'elaborazione.

## Questa guida dell'utente e riferimento

Per iniziare il più rapidamente possibile, cominciare dalla sezione *Avvio rapido* [↗](#) (pag. 5). Per maggiori informazioni sulla configurazione in background e dettagliata, leggere *Installazione e configurazione* [↗](#) (pag. 9). Per comprendere come eseguire le tipiche operazioni di elaborazione, consultare *Funzionamento di base* [↗](#) (pag. 15). Per informazioni dettagliate sulla sintassi e semantica di integrazione di WPS con R, studiare la sezione *Riferimento verso la fine*.

# Avvio rapido

Per iniziare rapidamente, è necessario avere esperienza con il linguaggio di SAS ed R.

Prima di iniziare, verificare che:

- La versione 3.1 di WPS o versioni successive, ed R (la versione 2.15.x o la versione 3.x) sono installate nel sistema. Questa breve esercitazione presuppone che sia presente un'installazione di WPS per Windows, ma le implementazioni in altri sistemi operativi sono molto simili.
- L'installazione di R include l'API della raccolta condivisa. È inclusa nelle installazioni predefinite in Windows e Mac OS X, ma non è inclusa per impostazione predefinita su altre piattaforme, quale Linux. Consultare la sezione *Installazione e configurazione* [↗](#) (pag. 9) per maggiori informazioni.

## Avvertenza:

La procedura R non funzionerà correttamente se l'installazione di R non include l'API della raccolta condivisa.

- La generazione e la gestione automatiche di ODS HTML vengono abilitate all'interno di WPS Workbench. Impostarle tramite **Finestra > Preferenze > WPS > Opzioni per risultati**.

Il seguente programma di esempio:

- Specifica il percorso dell'installazione di R.
  - Crea un set di dati di WPS denominato `origine`.
  - Trasmette il set di dati di origine all'ambiente di R.
    - Stampa il set di dati dall'interno del linguaggio R.
    - Esegue una semplice elaborazione e tracciamento di dati di origine con il linguaggio R.
    - Crea un semplice set di dati denominato `x` in linguaggio R.
  - All'interno dell'ambiente di WPS, recupera il set di dati `x` dall'ambiente di R.
  - Stampa il set di dati `x` mediante il linguaggio di SAS.
1. Avviare WPS e creare un nuovo programma (anche denominato *script*).

2. Copiare e incollare il seguente codice nel file e assicurarsi che il percorso nella prima riga punti all'installazione di R locale, per esempio:

```
options SET=R_HOME "C:\Program Files\R\R-2.15.3";


data source;
  do x=1 to 10;
    y=ranuni(-1);
  output;
  end;

PROC R;
  export data=source;
  submit;
    str(source)
    print(source)

    model <- lm(source$y ~ source$x)
    print(model)
    par(mfrow=c(2, 2))
    plot(model)

    x <- (1:10)
  endsubmit;
import R=x;
run;

proc print data=x;
run;
```

3. Salvare il programma ed eseguirlo facendo clic sull'icona della barra degli strumenti **Esegui** .
4. Esaminare il registro generato e l'output ODS.

L'echo di origine R standard viene indirizzato al file di registro di WPS:


```
NOTE: Submitting statements to R:

>   str(source)
>   print(source)
>
>   model <- lm(source$y ~ source$x)
>   print(model)
>   par(mfrow=c(2, 2))
>   plot(model)
>
>   x <- (1:10)

NOTE: Processing of R statements complete
```

L'output HTML per il server locale contiene l'output dalle istruzioni `print` e `plot`:

```
'data.frame': 10 obs. of 2 variables:
 $ x: num 1 2 3 4 5 6 7 8 9 10
 $ y: num 0.319 0.0167 0.4796 0.0208 0.6477 ...
  x      y
1  1 0.31896910
2  2 0.01672716
3  3 0.47964911
4  4 0.02080863
5  5 0.64773289
6  6 0.48497394
7  7 0.30747265
8  8 0.70942907
9  9 0.42861953
10 10 0.62938010
Call:
lm(formula = source$y ~ source$x)
Coefficients:
(Intercept)      source$x
  0.15361         0.04559
```



Gli ambienti WPS ed R ora sono integrati ed è possibile esplorare la completa gamma di possibilità disponibili, avendo l'accesso simultaneo ad entrambi i linguaggi. Il concetto fondamentale risiede nel fatto che è possibile collocare il codice R tra le istruzioni `submit` e `endsubmit`, impiegando altre istruzioni di `PROC R` in modo da gestire il trasferimento dei dati tra i due ambienti.



# Installazione e configurazione

Questa sezione contempla l'installazione e la configurazione dell'ambiente di WPS ed R.

## Software necessario

La procedura R è in dotazione con la versione 3.1 di WPS e versioni successive, ed è possibile utilizzare sia la versione a 32 che a 64 bit. Non è necessario installare altri moduli e non vi sono requisiti specifici per la gestione delle licenze dato che la procedura R fa parte del software WPS di base concesso in licenza. Tuttavia, il software WPS non viene spedito con una copia di R e, per usare la procedura R, sarà necessaria un'installazione di R nel computer.

### Nota:

Al momento della scrittura, le due principali versioni di R in uso sono 2.15.x e 3.x. Il software WPS si può utilizzare con ciascuna versione.

Consultare le *Note specifiche della piattaforma* [↗](#) (pag. 12) per informazioni sulle modalità di installazione di R in diverse piattaforme.

## Impostazione della variabile di ambiente *R\_HOME*

Per poter trovare la versione installata di R, WPS richiede l'impostazione della variabile di ambiente *R\_HOME*.

Questa operazione non è necessaria nella piattaforma Windows se viene utilizzato il percorso di installazione di R predefinito, dato che WPS può recuperare le informazioni dal registro di sistema di Windows. Se si esegue WPS con R su una piattaforma Unix o Linux, sarà necessario impostare la variabile *R\_HOME* in modo che punti alla cartella contenente `libr.so`. Se vi sono più istanze di R installate, *R\_HOME* deve puntare alla cartella contenente `libr.dll` su piattaforme Windows o `libr.so` su piattaforme UNIX o Linux.

I modi con cui è possibile impostare la variabile di ambiente *R\_HOME* dipendono dall'installazione specifica e dallo scenario di utilizzo:

Installazione e scenario di utilizzo	Istruzione OPTIONS	Variabile di ambiente di avvio del server	Script del profilo shell UNIX	Variabile a livello di sistema*
Workstation autonoma	Sì	Sì	Sì	Sì

Installazione e scenario di utilizzo	Istruzione OPTIONS	Variabile di ambiente di avvio del server	Script del profilo shell UNIX	Variabile a livello di sistema*
Server autonomo	Sì	Sì	Sì	Sì
Connessione desktop remoto al server	Sì	Sì	Sì	Sì
Collegamento di WPS tra workstation e server	Sì	N/D	Sì	Sì

\* Consultare la documentazione del sistema operativo specifico per maggiori informazioni sulle modalità di impostazione di una variabile a livello di sistema.

Quando si imposta la variabile di ambiente *R\_HOME* esternamente nella piattaforma di propria scelta, è possibile verificare la variabile tramite un programma SAS contenente la seguente istruzione:

```
%let EnvVar = %sysget( R_HOME ); %put "R_HOME is set to &EnvVar";
```

## Impostazione di *R\_HOME* tramite l'istruzione OPTIONS (tutti i sistemi operativi)

Il modo più semplice per impostare la variabile di ambiente *R\_HOME* è di utilizzare l'istruzione OPTIONS in un programma WPS.

1. Richiamare il seguente codice prima della procedura R, sostituendo il percorso nella propria installazione di R:

```
options SET=R_HOME "C:\Program Files\R\R-2.15.3";
```

Imposterà la variabile di ambiente *R\_HOME* per il resto della sessione di WPS. Un importante vantaggio di questo metodo è che consente di usare più versioni di R in un'unica sessione di WPS:

```
OPTIONS SET=R_HOME "C:\Program Files\R\R-2.15.3";
PROC R;
  SUBMIT;
    R.version
  ENDSUBMIT;
RUN;

OPTIONS SET=R_HOME "D:\Program Files\R\R-3.0.3";
PROC R;
  SUBMIT;
    R.version
  ENDSUBMIT;
RUN;
```

2. L'output da questo programma è simile a quello seguente:

```
platform      x86_64-w64-mingw32
arch          x86_64
os            mingw32
system        x86_64, mingw32
status
major         2
minor         150.3
year          2013
month         03
day           01
svn rev       62090
language      R
version.string R version 2.15.3 (2013-03-01)
nickname      Security Blanket
```

e

```
platform      x86_64-w64-mingw32
arch          x86_64
os            mingw32
system        x86_64, mingw32
status
major         3
minor         0.3
year          2014
month         03
day           06
svn rev       6512
language      R
version.string R version 3.0.3 (2014-03-06)
nickname      Warm Puppy
```

## Impostazione di *R\_HOME* tramite una variabile di ambiente di avvio del server WPS (tutti i sistemi operativi)

Un metodo alternativo per specificare il percorso dell'installazione R è quello di impostare la variabile di ambiente *R\_HOME* con l'opzione **Avvio** del server WPS, laddove WPS ed R sono installati.

1. Assicurarsi di aver eseguito l'accesso alla workstation o al server fisico in cui sia WPS che R sono installati.
2. Avviare WPS Workbench.
3. Fare clic con il pulsante destro del mouse sul server WPS locale nella scheda **Esplora server WPS** e selezionare **Proprietà**.

Viene visualizzata la finestra di dialogo **Proprietà per il Server locale**.

4. Assicurarsi che l'elemento **Variabili di ambiente** sia selezionato sul lato sinistro di questa finestra di dialogo, in **Avvio**.

5. Fare clic sul pulsante **Nuovo**, aggiungere la nuova variabile di ambiente e fare clic su **OK** nella finestra di dialogo **Proprietà**.
6. Viene richiesto di riavviare il server per applicare le modifiche.

In pochi secondi, il server sarà stato riavviato ed **Esplora server WPS** indicherà che un insieme di **Raccolte** e **Fileref** sono nuovamente associati ad esso. WPS utilizzerà la nuova variabile di ambiente quando deve individuare l'installazione di R locale.

## Impostazione di *R\_HOME* in uno script del profilo shell UNIX

Un'opzione finale per le piattaforme UNIX è quella di impostare il valore di *R\_HOME* in uno script del profilo shell.

Per esempio, si potrebbe impostare la variabile di ambiente in un file `~/ .bashrc` dell'utente.

---

**Nota:**

A causa del modo con cui le applicazioni vengono avviate in Mac OS X, questa tecnica non ha alcun effetto sull'ambiente per WPS Workbench, all'avvio, pertanto non funzionerà su quella piattaforma.

---

## Note specifiche della piattaforma

### Windows

È necessario installare la distribuzione binaria di Windows standard dal sito Web del progetto R.

Il pacchetto che è possibile scaricare in <http://www.r-project.org/> include versioni di R sia a 32 che a 64 bit, quindi lo stesso bundle funziona per le versioni del software WPS sia a 32 che a 64 bit.

Per impostazione predefinita, l'installazione di R salva il percorso di installazione nel registro di sistema di Windows, che il luogo in cui WPS cerca di trovare la versione attualmente installata. Questa sarà la versione di R installata più recentemente e non è necessaria nessuna configurazione specifica di WPS per trovarla. È possibile tuttavia impostare la variabile di ambiente *R\_HOME* mediante un'istruzione `OPTIONS` (pag. 10) oppure come opzione di avvio di WPS Server (pag. 11) se si desidera utilizzare una versione specifica di R.

## Piattaforme UNIX o Linux

La raccolta condivisa necessaria non è inclusa con la distribuzione binaria di R per piattaforme UNIX per impostazione predefinita, e non è attualmente possibile scaricare una idonea distribuzione binaria precompilata di R dal sito Web del progetto R.

Su piattaforme UNIX o Linux, è necessario compilare R dall'origine in modo che includa la raccolta condivisa richiesta oppure installare R tramite il sistema di gestione dei pacchetti di sistemi.

È necessario un minimo insieme di raccolte preinstallate prima di poter compilare R dal codice sorgente. Sono equivalenti a *build essentials* oltre a JDK in Ubuntu. Per maggiori informazioni, consultare la documentazione di R all'indirizzo <http://www.r-project.org/>.

Per installare R dal codice sorgente:

1. Scaricare il codice sorgente di R da <http://www.r-project.org/>.
2. Decomprimere il bundle del codice sorgente.

```
tar -xzf R-<version>.tar.gz
```

Crea una directory denominata R-<versione>.

3. Passare alla nuova directory e configurare il processo di compilazione.

```
./configure --enable-R-shlib --prefix=$HOME/R
```

Assicurarsi di utilizzare l'opzione `--enable-R-shlib` quando si esegue la configurazione poiché compila la raccolta condivisa `libr.so`.

4. Compilare e collegare il software R.

```
make
```

5. Installare il software R.

```
make install
```

Una volta installato, impostare la variabile `R_HOME` in modo che punti alla cartella contenente il file `libr.so`.

Se si è installato R tramite il sistema di gestione dei pacchetti di sistemi, è necessario trovare `libr.so` nel sistema e impostare `R_HOME` in modo che punti a quella cartella.

Ulteriori istruzioni per l'installazione di R dall'origine sono disponibili nella documentazione di R.

## macOS

È possibile installare direttamente la distribuzione binaria di R dal sito Web del progetto R.

Per usare R con WPS, è necessario impostare la variabile `R_HOME` in modo che punti alla directory di installazione contenente la raccolta condivisa `libr.so`. La corretta impostazione è: `/Raccolta/Framework/R.framework/Risorse`. Questa utilizzerà la versione predefinita di R.

---

**Nota:**

A causa del modo in cui le applicazioni vengono avviate in macOS, non è possibile impostare `R_HOME` in uno script del profilo shell.

---

Per utilizzare una versione specifica di R, è possibile modificare adeguatamente l'impostazione. Per esempio, `/Library/Frameworks/R.framework/Versions/3.0/Resources`

# Funzionamento di base

## Test dell'installazione e configurazione di WPS e R

Per effettuare il test dell'installazione e configurazione di WPS e R, è possibile creare ed eseguire un programma breve. Servirà inoltre da introduzione alla sintassi di PROC R.

1. Creare un nuovo file di programma, quindi incollare e salvare il seguente codice:

```
proc r;  
  submit;  
    x <- (1:10)  
    print(x)  
  endsubmit;  
run;
```

2. Eseguire il programma facendo clic sull'icona della barra degli strumenti **Esegui** .

Al termine dell'esecuzione del programma, il registro del server locale dovrebbe contenere gli elementi mostrati nel seguente esempio:

```
4      proc r;  
5      submit;  
6          x <- (1:10)  
7          print(x)  
8          endsubmit;  
NOTE: Using R version 3.0.3 (2014-03-06) from D:\Program Files\R\R-3.0.3  
  
NOTE: Submitting statements to R:  
  
>      x <- (1:10)  
>      print(x)  
  
NOTE: Processing of R statements complete  
  
9      run;  
NOTE: Procedure r step took :  
      real time : 0.406  
      cpu time  : 0.015
```

Il software WPS stampa il numero e il percorso della versione di R in uso. Questo si verificherà per ogni chiamata di PROC R. L'echo di origine di R standard è anche incluso nel registro di esecuzione di WPS, e l'output HTML del server locale contiene l'output dell'istruzione `print(x)`:

```
[1] 1 2 3 4 5 6 7 8 9 10
```

# Trasferimento dei dati tra WPS ed R

Le istruzioni `EXPORT` e `IMPORT` consentono il trasferimento di dati da e verso WPS ed R.

## Uso dell'istruzione `EXPORT`


L'istruzione `EXPORT` della procedura R viene utilizzato per trasmettere i dati dall'ambiente di WPS all'ambiente di R.

Questo succede normalmente quando è necessario mettere a disposizione in un contesto R alcuni dati generati in passaggi precedenti con il linguaggio di SAS.

1. Creare un nuovo file di programma, quindi incollare il seguente codice e salvare il file:

```
data source;
  do x=1 to 10;
    y=ranuni(-1);
    output;
  end;

proc r;
  export data=source;
  submit;
  str(source)
  print(source)
  endsubmit;
run;
```

2. Eseguire il programma facendo clic sull'icona della barra degli strumenti **Esegui**  ed esaminare l'output HTML.

Per impostazione predefinita, l'istruzione `EXPORT` crea un frame di dati con lo stesso nome del set di dati di WPS. Il frame di dati ha inoltre le stesse colonne del set di dati di WPS originale.

```
'data.frame': 10 obs. of 2 variables:
 $ x: num  1 2 3 4 5 6 7 8 9 10
 $ y: num  0.67611 0.15623 0.00499 0.76839 0.71394 ...

   x          y
1  1 0.676114940
2  2 0.156228246
3  3 0.004988594
4  4 0.768386004
5  5 0.713939278
6  6 0.004663629
7  7 0.714222821
8  8 0.453278570
9  9 0.613537647
10 10 0.072553599
```



## Uso dell'istruzione IMPORT


L'istruzione `IMPORT` della procedura R viene utilizzato per recuperare i dati dall'ambiente di R per tornare all'ambiente di WPS.

L'istruzione `IMPORT` accetta vettori, matrici e frame di dati di R, e li converte in adeguati set di dati di WPS.

1. Creare un nuovo file di programma, quindi incollare il seguente codice e salvare il file:

```
proc r;
  submit;
  x <- (1:10)
  endsubmit;
import R=x;

proc print data=x;
run;
```

2. Eseguire il programma facendo clic sull'icona della barra degli strumenti **Esegui**  ed esaminare l'output HTML.

Viene visualizzato di seguito l'output HTML dell'istruzione `proc print`.

---

**Nota:**

Il vettore di R denominato `x` è stato convertito in un set di dati denominato `x` nel linguaggio di SAS, contenente un'unica colonna, denominata anche `x`.

---

```
Obs x
1 1
2 2
3 3
4 4
5 5
6 6
7 7
8 8
9 9
10 10
```

## Uso della grafica di R


All'avvio di una sessione di R, WPS configura R in modo tale che i grafici generati con il dispositivo di grafica predefinito vengono acquisiti e inclusi nell'output ODS HTML standard della sessione di WPS.

Il seguente programma amplia l'esempio precedente fino ad includere l'analisi e la grafica per le regressioni lineari semplici.

1. Creare un nuovo file di programma, quindi incollare il seguente codice e salvare il file:





```
data source;
  do x=1 to 10;
    y=ranuni(-1);
    output;
  end;

PROC R;
  export data=source;
  submit;
  model <- lm(source$y ~ source$x)
  print(model)
  par(mfrow=c(2, 2))
  plot(model)
endsubmit;
run;
```

2. Eseguire il programma facendo clic sull'icona della barra degli strumenti **Esegui**  ed esaminare l'output HTML.

L'output include i risultati di R stampati unitamente ad un grafico semplice generato dall'interno della sessione di R e indirizzato all'output di WPS.

```
Call:
lm(formula = source$y ~ source$x)
Coefficients:
(Intercept)      source$x
    0.5344         0.0241
```



## Uso di ulteriori pacchetti di R

Per utilizzare ulteriori pacchetti che non sono inclusi nella normale installazione di R, si consiglia di usare l'ambiente della riga di comando di R interattiva per eseguire l'installazione e per controllare il

funzionamento di base di questi pacchetti, prima di provare ad utilizzarli all'interno della procedura R di WPS.

---

**Nota:**

Una sessione di R avviata da WPS eredita l'elenco delle variabili di ambiente dal processo di WPS. Quando si esegue da WPS Workbench nel computer, il processo del server WPS locale a sua volta eredita il suo elenco di variabili di ambiente dal processo di Workbench. Se il software di terzi viene installato per l'uso all'interno di R che richiede, ad esempio, ulteriori voci nella variabile di ambiente PATH, sarà necessario riavviare WPS Workbench per applicare le modifiche. Il solo riavvio del server locale in Workbench non è sufficiente per applicare le modifiche alle variabili di ambiente.

---

# Riferimento

Le definizioni dei diagrammi sintattici sono notazioni che aiutano a spiegare la sintassi dei linguaggi di programmazione, e sono utilizzate nella presente guida per descrivere la sintassi del linguaggio.

## Come leggere gli diagrammi sintattici

I diagrammi sintattici sono una notazione di sintassi grafica che accompagna le strutture del linguaggio significative, quali procedure, istruzioni e così via.

La descrizione di ciascun concetto linguistico comincia con la diagramma sintattico.

### Immissione di testo

Il testo che è necessario immettere esattamente nel modo in cui è visualizzato, viene mostrato con un carattere macchina da scrivere:




Questo esempio descrive un frammento di sintassi in cui la parola chiave `OUTPUT` viene seguita da un carattere punto e virgola: `;`. La forma del diagramma sintattico è:

Generalmente le maiuscole/minuscole del testo non sono importanti, ma in questo contesto, è consuetudine utilizzare le maiuscole per le parole chiave.

### Elementi segnaposto


I segnaposto, che bisogna sostituire con testo pertinente e dipendente dal contesto, sono rappresentati in un carattere minuscolo e corsivo:



Qui, è necessario immettere la parola chiave `OUTPUT` letteralmente, ma bisogna sostituire il *nome-del-set-di-dati* con qualcosa di appropriato per il programma, in questo caso, il nome del set di dati a cui aggiungere una osservazione.


### Opzionalità

Quando gli elementi sono opzionali, vengono visualizzati su un ramo sotto la linea principale nei diagrammi sintattici. L'opzionalità è rappresentata da un percorso alternativo senza ostacoli attraverso il diagramma:




## Ripetizione

Nei diagrammi sintattici, la ripetizione viene raffigurata con un cappio di ritorno che specifica optionalmente il separatore che bisogna collocare tra istanze multiple.




In alto, è necessario immettere la parola chiave `OUTPUT` letteralmente e farla seguire da una o più ripetizioni del `nome-del-set-di-dati`; in questo caso, non è stato necessario nessun separatore, tranne uno spazio.

L'esempio in basso dimostra l'uso di un separatore.



## Opzioni


Nei diagrammi sintattici, l'opzione è illustrata mediante diversi rami paralleli.




Nell'esempio illustrato in alto, è necessario immettere la parola chiave `GETNAMES` letteralmente e poi la parola chiave `YES` o la parola chiave `NO`.

## Frammenti

Quando la sintassi è troppo complicata per rientrare in una definizione, si potrebbe suddividere in frammenti:



## option



In alto, l'intera sintassi è suddivisa in frammenti di diagrammi sintattici. Il primo indica che `PROC PRINT` deve essere seguito da una o più istanze di un'opzione, ciascuna delle quali deve aderire alla sintassi fornita nel secondo diagramma.


# Procedura R

La procedura R viene richiamata dall'istruzione `PROC R` e consente l'esecuzione del codice programma scritto in R.


## Istruzioni supportate

- `PROC R` [↗](#) (pag. 23)
- `ASSIGN` [↗](#) (pag. 24)
- `EXECUTE` [↗](#) (pag. 26)
- `EXPORT` [↗](#) (pag. 26)
- `IMPORT` [↗](#) (pag. 28)
- `LOAD` [↗](#) (pag. 31)
- `SAVE` [↗](#) (pag. 32)
- `SUBMIT` [↗](#) (pag. 33)

## PROC R



## option



## GMTOFFSET="+/-HH:MM"

Imposta l'offset su GMT da applicare quando si spostano i valori di data-ora tra WPS e R tramite le istruzioni `ASSIGN`, `EXPORT` o `IMPORT`. I valori di data-ora in WPS non hanno un fuso orario, mentre i valori di data-ora in R sono rappresentati in TUC (Tempo Universale Coordinato) con un fuso orario associato.

## TIMESASCHRON

Controlla se i valori dell'ora sono rappresentati in R con la classe `chron`. Per impostazione predefinita, i valori dell'ora sono rappresentati in R tramite un conteggio di secondi dalla mezzanotte, ma facoltativamente è possibile utilizzare il pacchetto `chron`. Tuttavia, questo pacchetto non fa parte dell'installazione di R standard.

L'opzione influisce sulle istruzioni `ASSIGN` ed `EXPORT`.

## LIB=raccolta-predefinita

Specifica la raccolta predefinita utilizzata dalle istruzioni `EXPORT`, `IMPORT`, `LOAD` e `SAVE`. Le descrizioni di tali istruzioni forniscono maggiori dettagli sull'impatto di questa opzione.

## KEEP

Specifica che è necessario mantenere attiva l'applicazione di R alla fine del passaggio della procedura, e utilizzarla per le successive invocazioni di `PROC R`. Il comportamento predefinito consiste nell'interrompere l'applicazione di R alla fine della procedura, sebbene si possa alterare tramite l'opzione di sistema `RKEEP`.

## TERMINATE

Specifica che è necessario interrompere l'applicazione di R alla fine del passaggio della procedura. Questo è il comportamento predefinito, sebbene sia possibile cambiarlo tramite l'opzione di sistema `RKEEP`.

## TERM


`TERM` è un alias di `TERMINATE`.

## Un esempio di invocazione di PROC R

```
proc r;
  submit;
  R.version
  endsubmit;
run;
```


## ASSIGN

È possibile utilizzare l'istruzione `ASSIGN` per assegnare un valore ad un valore del vettore R.





## value



**Nota:**

Tutti i valori sul lato destro dell'attività deve essere dello stesso tipo. Per esempio, devono essere tutti numeri interi o tutte stringhe.

Il tipo di vettore R creato dipende dai tipi di valori forniti:

Tipi di valori forniti	Tipo di vettore R creato
Numeri interi	Numero intero
Virgola mobile	Numero reale
Data	Numero reale con una classe di <code>Date</code>
Datetime	Numero reale con una classe di <code>POSIXct</code> , configurate secondo l'opzione <code>GMTOFFSET</code> fornita all'istruzione <code>PROC R</code>
Ora	Numero reale con una classe di <code>times</code>
Stringa	Stringa

In genere, questa istruzione viene utilizzata per trasferire i parametri in un programma R. Spesso i valori a destra dell'assegnazione vengono generati tramite l'espansione della variabile della macro o l'esecuzione della macro. Con questa funzionalità, si riduce molto la necessità di eseguire l'espansione della macro all'interno dello stesso blocco `SUBMIT`.

Il nome dell'oggetto di R viene specificato come un normale identificatore nel linguaggio di WPS. Le maiuscole/minuscole vengono mantenute alla creazione dell'oggetto di R. Se necessario, è possibile utilizzare un valore letterale del nome (come in `"r.object.name"`) per creare oggetti di R che hanno nomi che non sarebbero altrimenti validi nel linguaggio di SAS.

## Un esempio di assegnazione di un valore ad un oggetto di R

```
%let parm=15;
proc r;
  assign parm=&parm;
  submit;
  x<-sample(1:3, parm, replace=TRUE)
  print(x);
  endsubmit;
run;
```

## EXECUTE

L'istruzione `EXECUTE` della procedura R consente l'esecuzione di un programma R archiviato in un file.



### Nota:

Utilizzando l'istruzione `EXECUTE` è un'alternativa all'uso dell'istruzione `SUBMIT`. Consente la collocazione del codice di R in un file separato. Ciò è utile perché consente l'esecuzione dello stesso codice programma direttamente in un ambiente di R interattivo. Laddove è pertinente, i nomi di percorso relativo vengono risolti in modo relativo alla directory corrente del processo di WPS.

## Un esempio di esecuzione di un programma R archiviato in un file

Contenuto del file di origine `model.r`:

```

model <- lm(source$y ~ source$x)
print(model)
par(mfrow=c(2, 2))
plot(model)
  
```

Esempio di invocazione della procedura R:


```

data source;
do x=1 to 10;
  y=ranuni(-1);
  output;
end;


PROC R;
  export data=source;
  execute "model.r";
run;
  
```

## EXPORT

L'istruzione `EXPORT` della procedura R crea un frame di dati R da un set di dati di WPS.



### option



## DATA

Specifica il nome del set di dati da esportare, insieme alle opzioni del set di dati necessario. Questa opzione è obbligatoria.

## R o FRAME

Specifica facoltativamente il nome da assegnare all'oggetto nell'ambiente di R. Se viene omissso, il nome dell'oggetto di R viene preso dal nome del membro del set di dati.

È possibile specificare normalmente le opzioni del set di dati nel set di dati di input, ma potrebbe essere utile applicare una clausola `WHERE` o un elenco `DROP` al set di dati prima di esportarlo a R.

### Nota:

L'applicazione della clausola `WHERE`, o l'esportazione di un set di dati da una raccolta o vista sequenziale (in altre parole, quando il numero di osservazioni nel set di dati non è noto) richiede ulteriori risorse, dato che è necessario prima inviare allo spooler il set di dati per calcolare il numero esatto di osservazioni prima che si possa creare il frame di dati di R.

Esistono due tipi di variabile in un set di dati di WPS: *numerico* e *carattere*. Inoltre, una colonna numerica può avere un formato associato ad essa che si può utilizzare per immettere ulteriori informazioni. Il tipo di vettore R creato è come riportato di seguito:

Tipo di variabile di WPS	Tipo di vettore R creato
Carattere	Stringa standard
Variabile numerica con formato di data applicato	Assegnata una classe di <code>Data</code> al vettore R reale
Variabile numerica con un formato di datetime applicato	Un vettore R reale verrà creato e verrà assegnata una classe di <code>POSIXct</code> . I valori di tale classe rappresentano un conteggio di secondi dal 1° gennaio 1970 (Tempo Universale Coordinato). I valori vengono regolati in base all'opzione <code>GMTOFFSET</code> trasferita all'invocazione <code>PROC R</code> per tenere conto del fatto che i valori di datetime nel linguaggio di SAS sono l'ora locale, mentre i valori della classe <code>POSIXct</code> in R devono essere in TUC.
Variabile numerica con un formato di ora applicato	Esistono due opzioni basate sul fatto che l'opzione <code>TIMESASCHRON</code> viene specificata dall'invocazione <code>PROC R</code> . Per impostazione predefinita, un normale vettore R reale viene creato e nessuna classe speciale viene assegnata al vettore. Tuttavia, se viene specificata l'opzione <code>TIMESASCHRON</code> , allora si assegna al vettore una classe di <code>ore</code> . Il pacchetto <code>chron</code> di R (che fornisce la classe <code>times</code> ) non fa parte di un'installazione standard di R, ma fornisce le utilità per la gestione dei valori dell'ora del giorno.
Altro valori numerici	Viene creato un normale vettore R reale e non viene assegnata nessuna classe specifica.

Quando si trasferiscono i valori numerici da WPS a R, l'istruzione `EXPORT` interpreta i valori mancanti specifici `.I` e `.M`, e di conseguenza crea i valori di R `Inf` e `-Inf`. Qualsiasi altro valore mancante viene trasferito ad R come il valore `NaN`.

## Un esempio di esportazione dei dati da WPS a R

Questo esempio crea un set di dati contenente due colonne numeriche e lo esporta in R.

```
data source;
do x=1 to 10;
  y=ranuni(-1);
  output;
end;


proc r;
  export data=source;
  submit;
  str(source)
endsubmit;
run;
```

Il frame di dati risultante:


```
'data.frame': 10 obs. of 2 variables:
 $ x: num 1 2 3 4 5 6 7 8 9 10
 $ y: num 0.371 0.924 0.59 0.434 0.962 ...
```

## IMPORT

L'istruzione `IMPORT` della procedura R crea un set di dati di WPS da un oggetto di R.



### option



### DATA

Specifica facoltativamente il percorso in cui è necessario salvare il set di dati. Questo può includere le opzioni del set di dati. Se viene ommesso, il set di dati viene salvato nella raccolta predefinita (`WORK` o la raccolta denominata nell'opzione `LIB=` nell'invocazione di `PROC R`).

## R o FRAME

Specifica il nome dell'oggetto di R da importare che deve essere sotto forma di un identificatore, non un valore letterale di una stringa tra virgolette. È possibile utilizzare il valore letterale di un nome qui per specificare un nome che non è normalmente valido per un identificatore nel linguaggio di SAS. Questa opzione è obbligatoria.

## Regole della conversione di IMPORT

È possibile importare qualsiasi oggetto che si può forzare in un frame di dati tramite la funzione di R `as.data.frame`.

Se non è possibile forzare l'oggetto di R specificato in un frame di dati, viene prodotto un errore. WPS può importare le colonne che hanno i tipi di R logico, numero intero, numero reale e carattere. Inoltre, può importare i fattori. Le colonne con il tipo logico, numero intero o reale vengono convertiti in colonne numeriche nel set di dati di WPS. Le colonne con il tipo carattere e fattori vengono convertite in colonne di stringa. Le seguenti note offrono maggiori dettagli su queste regole di conversione.

### Valori logici

I valori dei vettori di tipo logico vengono convertiti come riportato di seguito:

Valore di R	Valore di WPS
TRUE	1
FALSE	0
NA	.

### Valori dei numeri interi

Lo specifico valore NA in R, rappresentato in R come il minimo valore di numero intero (-2147483648) viene convertito nel valore mancante del linguaggio di SAS.

### Valori dei numeri reali

Esistono tre specifici valori numerici reali nel linguaggio R: NA, NaN e Inf. In R, NA viene utilizzato per rappresentare un valore assente (non disponibile), Inf denota l'infinito (dividere per zero per esempio) e NaN rappresenta un non-numero (il risultato di 0/0 per esempio). Tali valori vengono convertiti come riportato di seguito:

Valore di R	Valore di WPS
NA	.
NaN	.
+Inf	.I

Valore di R	Valore di WPS
-Inf	.M

## Valori di data

Alle colonne di numeri interi o reali che hanno una classe R di `Data` viene applicata una elaborazione specifica. Alla variabile nel set di dati di WPS viene assegnato un formato di `DATE9` e, dopo l'importazione, i valori vengono regolati in modo da tenere conto della differenza nel periodo utilizzato in R e WPS. I valori della classe `Data` in R vengono rappresentati come un conteggio dei giorni dal 1° gennaio 1970, mentre nel linguaggio di SAS, il periodo è 1° gennaio 1960.

## Valori di data-ora

WPS applicherà una gestione particolare alle colonne di numeri reali che hanno la classe `POSIXct`. I valori di tale classe in R rappresentano un conteggio di secondi dal 1° gennaio 1970 in TUC. Quando vengono importate le colonne di questa classe, i valori vengono regolati in modo da tenere conto della differenza nel periodo tra SAS ed R. Alla colonna nel set di dati di WPS viene assegnato il formato `DATETIME19`. I valori vengono anche regolati in base al valore dell'opzione `GMTOFFSET` nell'invocazione `PROC R` per tenere in considerazione che i valori in R sono in TUC, mentre nel linguaggio di SAS, i valori di `datetime` sono nell'ora locale.

## Valori di ora

WPS applicherà una gestione particolare alle colonne di numeri reali che hanno la classe `ore`. Alla colonna nel set di dati di WPS viene assegnato il formato `TIME8`.

## Valori del carattere

WPS analizzerà i valori nella colonna dei caratteri per trovare il valore più lungo e assegnerà la lunghezza della colonna di WPS a quel valore. I valori singoli in una colonna di caratteri possono non essere disponibili (`NA`) in R, e questi saranno convertiti al valore del carattere mancante in WPS (cioè il valore sarà completamente formato da spazi). Non vi sarà alcuna differenza quindi tra i valori " " e `NA` se importati in WPS.

## Valori dei fattori


Un fattore in R è una forma specifica di colonna di numeri interi, laddove i valori dei numeri interi nella colonna vengono indicizzati in un elenco di valori univoci archiviato come attributo sulla colonna (vengono denominati i livelli in R). Se importati in WPS, questi vengono convertiti in variabili di caratteri nel set di dati. Alla colonna viene assegnata una lunghezza uguale alla stringa più lunga nell'elenco dei livelli.

## Un esempio di importazione dei dati da R a WPS


```
proc r;
  submit;
  x<-sample(1:3, 15, replace=TRUE)
  endsubmit;
import r=x data=demo_import;
run;
proc print data=demo_import;
run;
```

## LOAD

Le istruzioni `LOAD` e `SAVE` consentono la serializzazione e archiviazione temporanea o permanente degli oggetti di R in una raccolta di dati di WPS e, successivamente, la deserializzazione nella stessa sessione di WPS oppure in una successiva. L'istruzione `LOAD` deserializza un oggetto di R salvato in precedenza con l'istruzione `SAVE`.



### option



### CATALOG o CAT o C

Fornisce il percorso in cui l'oggetto di R verrà salvato. Questa opzione è obbligatoria. Se la raccolta non è specificata nell'opzione `CATALOG`, l'opzione `LIB=` assegna la raccolta predefinita nell'invocazione di `PROC R` o, altrimenti, normalmente viene utilizzata la raccolta `USER` o `WORK`.

### R o FRAME

È possibile specificare il nome dell'oggetto di R tramite la sintassi del valore letterale (per esempio, "r.object.name") se il nome dell'oggetto di R non soddisfa le normali regole per gli identificatori nel linguaggio di SAS. Le maiuscole/minuscole del nome vengono mantenute alla creazione dell'oggetto di R. Questa opzione è obbligatoria.

## Un esempio dell'uso dell'istruzione LOAD


```
proc r;
  load cat=catalog.entry r='target.object'n;
run;
```

## SAVE

Le istruzioni `LOAD` e `SAVE` consentono la serializzazione e archiviazione temporanea o permanente degli oggetti di R in una raccolta di dati di WPS e, successivamente, la deserializzazione nella stessa sessione di WPS oppure in una successiva. L'istruzione `SAVE` serializza un oggetto di R e lo archivia in una voce di un catalogo.



### option



## CATALOG o CAT o C

Questa opzione obbligatoria fornisce il percorso in cui l'oggetto di R verrà salvato. Se la raccolta non è specificata nell'opzione `CATALOG`, l'opzione `LIB=` assegna la raccolta predefinita nell'invocazione di `PROC R` o, altrimenti, normalmente viene utilizzata la raccolta `USER` o `WORK`.

## R o FRAME

Specifica il nome dell'oggetto di R da salvare. Si può specificare tramite la sintassi del valore letterale (per esempio, `"r.object.name"`) se il nome dell'oggetto di R non soddisfa le normali regole per gli identificatori nel linguaggio SAS. Poiché R è un linguaggio che distingue tra maiuscole e minuscole, le maiuscole/minuscole del nome devono corrispondere a quelle dell'oggetto di R. Questa opzione è obbligatoria.

## DESCRIPTION

Fornisce una stringa descrittiva che viene salvata nella voce del catalogo. Tale descrizione verrà visualizzata nell'output tramite l'istruzione `PROC CATALOG CONTENTS`.



**Nota:**


La voce del catalogo avrà un tipo di `OBJECT`.

## Un esempio di salvataggio di un oggetto di R ad un catalogo WPS

```
proc r;
  save cat=catalog.entry r='source.object'n;
run;
```

## SUBMIT

L'istruzione `SUBMIT` della procedura R consente l'esecuzione del codice programma in linea nel linguaggio R.



**Nota:**

Non è necessaria nessuna modifica al codice programma di R. È possibile copiare e incollare il normale codice programma in linguaggio R, circondarlo con istruzioni `SUBMIT` ed `ENDSUBMIT`, e richiamarlo dall'interno della procedura R.

Il codice sorgente R deve iniziare in una nuova riga dopo l'istruzione `SUBMIT` e l'istruzione `ENDSUBMIT` deve apparire da sola all'inizio di una riga.

Possono esistere vari blocchi `SUBMIT` all'interno di una singola invocazione di `PROC R`. Ogni blocco `SUBMIT` viene eseguito non appena viene rilevato. È possibile interporre blocchi `SUBMIT` tra altre istruzioni se necessario.

## Elaborazione di macro

La natura del linguaggio R indica che le righe tra le istruzioni `SUBMIT` ed `ENDSUBMIT` vengono copiate letteralmente e trasferite all'ambiente di R. L'elaborazione di macro viene sospesa tra le istruzioni `SUBMIT` ed `ENDSUBMIT`. Esistono varie ragioni per cui si verifica:

- Il linguaggio R utilizza i caratteri `&` e `%` come parte della sintassi. Il tentativo di elaborare le macro del codice sorgente R può produrre l'equivocazione della sintassi di R corretta come per le invocazioni di macro o i riferimenti di WPS alle variabili delle macro in linguaggio di SAS.

- Il linguaggio R supporta commenti del tipo a fine riga, il contenuto dei quali potrebbe contenere ad esempio apostrofi senza corrispondenza. Questo potrebbe rendere difficile la creazione di token della sintassi R tramite le normali regole analitiche del linguaggio di SAS, che sarebbe necessaria per supportare l'elaborazione di macro del codice sorgente di R.

Inoltre, a causa del modo in cui funziona l'elaboratore di macro e il modo in cui gestisce le righe di codice sorgente, non è possibile generare un blocco `SUBMIT` tramite una macro. Vale a dire, un blocco `SUBMIT` non può apparire all'interno di una macro in linguaggio di SAS. Tuttavia, il blocco `SUBMIT` può apparire in un file identificato mediante un'istruzione `%INCLUDE`. Quindi, se è necessario per generare una `PROC R`, l'invocazione con l'elaboratore di macro, bisogna utilizzare l'istruzione `EXECUTE` o inserire il contenuto del blocco `SUBMIT` in un file separato che poi verrà identificato mediante un'istruzione `%INCLUDE`.

## Sostituzione di testo

Al posto dell'elaboratore di macro, viene fornita una semplice funzionalità di sostituzione di testo. Prima del trasferimento a R, le righe comprese tra `SUBMIT` e `ENDSUBMIT` si può applicare un numero limitato di sostituzioni. Le sostituzioni vengono fornite nell'istruzione `SUBMIT`. La sintassi per le sostituzioni è simile a quella per la normale sostituzione delle variabili delle macro. Tuttavia, non vi è una nuova analisi ed è permessa solo la semplice sintassi dello stile delle variabili delle macro di livello singolo.

```
&symbol
&symbol.
```

Persino con questa semplice sintassi e solo sostituendo i simboli esplicitamente elencati nell'istruzione `SUBMIT`, è possibile che si verifichino sostituzioni involontarie. Per evitarlo, si consiglia di scegliere nomi di simboli che siano diversi dai nomi di oggetti di R, ai quali WPS fa riferimento nel codice di R inviato. Il simbolo della "e" commerciale (& e &&) viene utilizzato da R come l'operatore logico *and*. Considerare il caso di un'istruzione `SUBMIT` contenente il seguente codice di R:

```
a>b&c<d
```

Se `c` viene definita come simbolo di sostituzione, produrrà la sostituzione involontaria di `c` dentro questa espressione. L'inclusione di uno spazio intorno alla "e" commerciale lo eviterà, come anche la scelta di nomi per i simboli di sostituzione che siano meno conflittuali con i nomi degli oggetti di R.

Non vi è alcuna sintassi di escape e non vi è alcun modo per evitare che un simbolo venga sostituito. Considerare il seguente caso:

```
proc r;
  submit Goodbye="hello";
    A <- "Hello&Goodbye"
  endsubmit;
run;
```

Non è possibile evitare la sostituzione in questo caso, oltre a scegliere un nome diverso per il simbolo, e la "e" commerciale non è un simbolo specifico dato che viene usato come operatore logico all'interno di R.


È possibile generare valori di sostituzione tramite le funzionalità del normale elaboratore di macro, come riportati nel seguente esempio:

```
proc r;  
  submit Goodbye="hello";  
  A<-"&sym"  
endsubmit;  
run;
```

## Esempio of invio del codice di R in linea

```
proc r;  
  submit;  
  x <- (1:10)  
  print(x)  
endsubmit;  
run;
```

# Ulteriore lettura

Un punto di partenza suggerito per ulteriori informazioni, inclusa la sintassi, la semantica e molti pacchetti di componenti aggiuntivi che ampliano l'utilità delle funzionalità di base del linguaggio, è il sito Web del progetto R all'indirizzo <http://www.r-project.org> .

# Avvisi legali

Copyright © 2002–2018 World Programming Limited.

Tutti i diritti riservati. Le presenti informazioni sono riservate e soggette a diritto d'autore. Non è possibile riprodurre o trasmettere nessuna parte di questa pubblicazione, in qualsiasi forma o con qualsiasi mezzo, elettronico o meccanico, inclusa la fotocopiatura, la registrazione o eventuali sistemi di archiviazione e recupero dati.

## Marchi commerciali

WPS e World Programming sono marchi registrati o marchi commerciali di World Programming Limited nell'Unione europea e altri paesi. (r) o ® indica un marchio comunitario.

SAS e tutti gli altri nomi di prodotti o servizi di SAS Institute Inc. sono marchi registrati o marchi commerciali di SAS Institute Inc. negli Stati Uniti e in altri paesi. ® indica la registrazione negli USA.

Tutti gli altri marchi commerciali sono proprietà dei rispettivi titolari.

## Avvisi generali

World Programming Limited non è associata in alcun modo a SAS Institute.

WPS non è SAS System.

Le frasi "SAS", "linguaggio SAS" e "linguaggio di SAS" utilizzate in questo documento si usano in riferimento al linguaggio di programmazione spesso denominato in uno dei suddetti modi.

Le frasi "programma", "programma SAS" e "programma in linguaggio SAS" utilizzate in questo sito Web si riferiscono a programmi scritti in linguaggio SAS, definiti anche "script", "script SAS" o "script in linguaggio SAS".

Le frasi "IML", "linguaggio IML", "sintassi IML", "Interactive Matrix Language" e "linguaggio di IML" utilizzate in questo documento si usano in riferimento al linguaggio di programmazione spesso denominato in uno dei suddetti modi.

WPS include software sviluppato da terzi. È possibile trovare maggiori informazioni nel file THANKS o acknowledgments.txt inclusi nell'installazione di WPS.